

3 Ways to Interactively Show

Top/Bottom 'X' Items

In Excel Dashboards



Using IF, SMALL, LARGE & Option Buttons



Using IF, RANK.EQ & Option Buttons



Using Pivot Tables

William Kiarie

Data Cycle Analytics

In the course of my career as an Excel Data Analyst and Business Intelligence Trainer, one of the intuitive techniques especially when it comes to creating dynamic and interactive Excel Dashboards is the ability to allow the dashboard user to select, say the top/bottom 3 items.

This could be the top 3 overspending projects for the case of a Project Manager, or bottom 5 selling products in a given region, or top 10 sales reps by revenue in a given quarter, and so forth depending on what industry one is.

But not many people are able to do this in Excel, and even those who attempt to, they are stuck on how to they can allow the users to choose what view of the data they want. This article is written to address this specific challenge.

This eBook revolves around these three approaches:

1. Using Pivot Tables
2. Using LARGE & SMALL Functions
3. Using the RANK Function

Let us dive right into our Excel workbook and start off!

“Interactive Dashboards in Excel should allow the user to choose what view they want, say top 3 or bottom 3. #ExcelDashboards

Method 1: Top/Bottom Views Using Pivot Tables

Pivot Tables offer a rich set of features that can make the life of an analyst easy. You can create very interactive reports by leveraging on these hidden features. One of these features is being able to scale a list of products and the revenues to the top or bottom 5.


This eBook will orient you on how to use pivot tables for top bottom views.




Be on the lookout for the release of my book **Pivot Driven Dashboards** that looks at a collection of these and more applications of pivot tables.

With the [Excel workbook you downloaded](#), select any given cell in the data and insert a pivot table from the **Insert Menu** on the Excel Ribbon. Drag the list of *SKUs* on the **Row labels** and *Amount* column in the **Values** section of the pivot report **Field List**. This will give you a report showing the sales revenues generated by each of the SKUs.

From the **filter handle** (see the figure below), select **Value Filters >>>Top 10**.



This will give you the dialog box shown below that allows you to set whether you want to see top or bottom items. You can also adjust from Top 10 to another number of your choice. In this example we have changed to Top 5.



As it is apparent from the figure above, you will need to repeat this process for bottom 5 SKUs. At the end of the day you will have *two sets of summaries*, one for top 5 and another for bottom 5 SKUs.

	A	B	C	D	E	F
1		Top 5 SKUs			Bottom 5 SKUs	
2						
3		Row Labels	SalesAmount		Row Labels	SalesAmount
4		Aamras 250 ml	\$1,236,749		Aamras 250 ml MRP 30	\$1,680
5		Lemon Mint 250 ml	\$1,327,353		JAMUN KALA KHATTA x 24	\$912
6		Mango Strawberry 250 ml	\$1,270,362		LM 250ML MRP 50	\$3,840
7		Tropical Trip 250 ml	\$1,419,144		MS 250ML 24PCS 50	\$3,840
8		TT 250 ml	\$1,241,261		TT 250ML 24PCS 50/-	\$3,840
9		Grand Total	\$6,494,868		Grand Total	\$14,112

The challenge would arise if you are to show this in a dashboard where the user needs to toggle (using say, *Option Buttons* or a *Custom Slicer*) between the top and bottom.

The technique applicable in that case would involve the use of a combination of the **CHOOSE** function, **named ranges**, and **Linked Picture** and/or the **Cameral tool** in Excel. This technique is a bit lengthy to include it as part of this eBook but I will discuss it in a separate article.

Method 2: Using the LARGE and SMALL Functions

If you have used the MAX and MIN functions in Excel, then you know they return a single value representing the highest and the lowest value in a given range.

LARGE and SMALL functions, on the other hand, allow you to specify whether you want the 2nd, 3rd, 4th, largest/smallest value in the same range. As such, you therefore have to specify the range of values, as well as the 'Kth' element, where 'K' is any number 1, 2, 3, and so on.

To illustrate this, we start with the original summary of SKUs performance (the entire list of 27 SKUs) in the previous section.

Syntax:

=LARGE (Array, k) **or** **SMALL (Array, k)**

In a given empty column in your analysis sheet, type the numbers 1, 2, up to 5. These numbers are our 'k' parameter in the formula above. On an adjacent column, type the LARGE/SMALL function as shown below.

	H	I	J	K	L	M	N	O
1		LARGE & SMALL functions						
2								
3		Row Labels	SalesAmount		Pos	SKU	SalesAmount	
4		TT x 24 SPH 250ML30	\$357,540		1		=LARGE(\$J\$4:\$J\$30,\$L4)	
5		TT 250ML 24PCS 50/-	\$3,840		2		LARGE(array, k) 3	
6		TT 250 ml	\$1,241,261		3		1,270,362	
7		Tropical Trip 250 ml	\$1,419,144		4		1,241,261	
8		MS x 24 SPH 250ML30	\$450,886		5		1,236,749	
9		MS 250ML 24PCS 50	\$3,840					

Notice that to get the *bottom 5 list*, you just change the function used from LARGE to SMALL. Other parameters aren't changed.

To get the SKU name associated with the sales amount in cell **N4**, we shall insert the following formula in cell **M4**. I hope you are able to decipher the formula, if not, kindly leave a comment below.

Formula for SKU name: **=INDEX(\$I\$4:\$I\$30, MATCH(\$N4, \$J\$4:\$J\$30, 0))**

The output of these 2 formulas is as follows:


	K	L	M	N
2				
3		Pos	SKU	SalesAmount
4		1	Tropical Trip 250 ml	1,419,144
5		2	Lemon Mint 250 ml	1,327,353
6		3	Mango Strawberry 250 ml	1,270,362
7		4	TT 250 ml	1,241,261
8		5	Aamras 250 ml	1,236,749

To create a dynamic list that would work with our Excel Dashboard, we need to tweak the LARGE function used above by using an **IF function** and a value that the user controls using an **Option Button**.


We start by inserting an option button from the **Developer menu** on the Excel ribbon. If the Developer menu/tab isn't available for you, you need to enable it by right clicking anywhere on the ribbon, choose **Customize the Ribbon...** and tick the Developer menu on the dialog box that appears.

Insert an Option Button

On the Developer tab, **Controls group**, choose **Insert** and select the **Option Button** icon as shown below.




Draw a reasonable size of the option button. Right click on the option button created and select **Edit text**. This will allow you to delete the default text that comes with the option button -once deleted, type TOP 5 SKUs. Next, copy this option button and once you paste, edit the text to read BOTTOM 5 SKUs as follows:



Next, right click any of the two option buttons and choose **Format Control** from the short-cut menu that pops up. The **Format Control** dialog box will appear.

On the **Control tab**, choose the link cell that will store the index/number generated once a selection is made on the button. In this example, we select cell **\$L\$1**.



When you click OK, cell **\$L\$1** will not have a value until you make a selection on either of the option buttons. For Option Buttons, the number assigned for each is dependent on which one was inserted first. What we are interested with is this cell **\$L\$1** to help us build a dynamic LARGE/SMALL formula.

For readability of formulas, we have named cell **\$L\$1** as **selOption** from the Name Box. We adjust the formula in cell **N4** as follows:

=IF (selOption=1, LARGE (\$J\$4:\$J\$30, \$L4), SMALL(\$J\$4:\$J\$30, \$L4))

This will give you the flexibility to display top performing or bottom 5 performing SKUs interactively.

“Most of the form controls (Option Buttons, Combo Boxes, Scroll Bars, etc.) generate a number representing the current selected item. This number helps you to do further analysis and achieve the desired interaction in your reports.

The challenge with this methodology as you may have noticed in the results of the formula, is the likeliness of having multiple SKUs with a similar amount.

When you have multiple SKUs with a similar amount, the INDEX formula that returns the name of the SKU will always return the same SKU name for the repeated amounts. This is where the first methodology of using pivot tables shines, since you are able to return all the products/SKUs with similar amount without a lot of trouble.

The results when the 'Bottom 5 SKUs' is selected:

	K	L	M	N
1		2	<input type="radio"/> TOP 5 SKUs	<input checked="" type="radio"/> BOTTOM 5 SKUs
2				
3		Pos	SKU	SalesAmount
4		1	JAMUN KALA KHATTA x 24	912
5		2	Aamras 250 ml MRP 30	1,680
6		3	LM 250ML MRP 50	3,840
7		4	LM 250ML MRP 50	3,840
8		5	LM 250ML MRP 50	3,840

The paragraph below explains how to address the above duplication challenge.

Handling Duplicates While Using LARGE/SMALL

LARGE/SMALL functions will return the repeated amounts correctly. The challenge is when you now use these amounts in the MATCH function that gives the row position for the INDEX function. The row number will be the same for all the repeated amounts thus making INDEX to return the same SKU name. To address this, we shall add a negligible amount, say 0.01, to subsequent amounts. We thus have a helper column with the adjusted figures using this formula:

$$=(\text{COUNTIFS}(\text{R\$4:R4}, \text{R4}) - 1) * 0.01 + \text{R4}$$

Method 3: Using the RANK Function

This is a close alternative of the second method discussed above. We still need to type the numbers 1, 2 up to 5.

This approach relies on assigning positions to the sales amounts relative to their magnitude using the RANK function. This function ranks a given number against a referenced range of numbers, in a specified order (ascending or descending), more like the ranking of students in their exam scores.

Syntax:

=RANK.EQ (number, ref, [order])

The 'number' parameter is the specific item/value you want to rank; The 'ref' is a list of numbers/values (the complete list) against which the 'number' is ranked; The 'order' parameter specifies the order of the ranking (it is 0 for descending order, and 1 for ascending order).

Thus, we shall enter the formula below to rank the amounts:

	Q	R	S	T	U
1	Using RANK function				
2					
3	Row Labels	SalesAmount	Rank		
4	TT x 24 SPH 250ML30	\$357,540	=RANK.EQ(\$R4,\$R\$4:\$R\$30,0)		
5	TT 250ML 24PCS 50/-	\$3,840	RANK.EQ(number, ref, [order])		
6	TT 250 ml	\$1,241,261	4		
7	Tropical Trip 250 ml	\$1,419,144	1		
8	MS x 24 SPH 250ML30	\$450,886	10		
9	MS 250ML 24PCS 50	\$3,840	23		
10	MS 250 ml	\$942,874	8		

Start by prepping where the top/bottom items will be displayed as follows:

	U	V	W	X
2				
3		Rank	SKU	SalesAmount
4		1		
5		2		
6		3		
7		4		
8		5		

In cell **W4**, we type the formula below and auto-fill it to the range **W4:X8**:

=INDEX(Q\$4:Q\$30, MATCH(\$V4, \$S\$4:\$S\$30, 0))

This formula populates the range **W4:X8** returning both the amount and the SKU names.

	U	V	W	X
2				
3		Rank	SKU	SalesAmount
4		1	Tropical Trip 250 ml	1,419,144
5		2	Lemon Mint 250 ml	1,327,353
6		3	Mango Strawberry 250 ml	1,270,362
7		4	TT 250 ml	1,241,261
8		5	Aamras 250 ml	1,236,749

At this juncture, this report only shows the top 5 SKUs as the order specified was 0(-zero), for descending order.

We therefore need to tweak the formula used to rank the sales amounts so that we can toggle between top and bottom as that is what we need.

Original formula in cell S4:

=RANK.EQ(\$R4, \$R\$4:\$R\$30, 0)

Corrected formula:

=RANK.EQ(\$R4, \$R\$4:\$R\$30, IF(selOption=1,0,1))

The adjusted part in the formula is as highlighted. This second formula takes the input from the *Option Buttons* earlier created.

Since the third parameter of the RANK function is either a 0 or 1, we use an IF function to check what the user has selected. If the user has selected “TOP 5 SKUs”, then we set the outcome of the IF function to a 0 (i.e. zero) to sort the sales amounts in a descending order (*highest value ranked as position 1*), otherwise ranks in an ascending order.

As you will notice, duplicate values are ranked using the same number. That is, if there is an amount repeated, say 4 times, these four values will share the same rank. This presents a problem when it comes to returning the top/bottom 5 items.

Dealing with duplicates in the ranks:

Notice that the duplicated sales amounts will share the same rank/position and Excel skips the next number to account for this duplication. We shall therefore tweak the RANK function above to add 0 (zero) to the rank value if the amount appears the first time, OR add 1 if it's the second occurrence, add 2 if it is the 3rd occurrence of the amount, and so on. The formula applied for this is COUNTIFS:

=RANK.EQ(\$R4,\$R\$4:\$R\$30,IF(selOption=1,0,1)) + COUNTIFS(R\$4:R4,R4) - 1

If the value appears on the list only once, there's no reason to change the original value. If the amount appears more than once, I want to add .01 to it based on the number of times that it's appeared on the list.

By locking row 4 in the COUNTIFS function above, and tricking Excel by counting within the same cell, R\$4:R4, we ensure that the COUNTIFS formula counts up-to the current row for which we have auto-filled the formula.

We are therefore guaranteed that the this COUNTIFS formula will return a 1 for the first occurrence of the amount, and therefore subtracting 1, makes the net effect to the RANK function to be zero. That is:

`=RANK.EQ($R4,$R$4:$R$30,IF(selOption=1,0,1)) + 1 - 1`

The second occurrence will be 2-1, the third occurrence will return 3-1, and so forth.

So What's Next for me?

Here we are at the end of this eBook and you might be wondering what's next! If you find this guide useful, please do share with anyone else in your circles. Colleagues at your place of work or in other organizations may be wondering how to do top bottom items in Excel! You can embed this guide on your blog/website, share it on a forum or anywhere you like, email it to a friend or use excerpts on your site, as long as you cite the source – DataCycleAnalytics.com

As for you I would love to hear how analysis in Excel is like for you. Post a comment at the blog and let me know if this guide has helped. I really hope it has.

About the Author



William Kiarie is the founder of datacycleanalytics.com, where he regularly publishes articles about Microsoft Excel, Power Pivot and Power BI and how to leverage on these tools to create meaningful management dashboards.

He provides consultancy and training on Business Intelligence (BI) with a bias on Microsoft technologies -MS Excel & MS Power BI.

Let us Connect:

